
1. Reading the dataset 

 

  use "F:\TA ECO310H1 2018FALL\Tutorial3\verboven_cars.dta", clear 

 

This directory depends on where you save the dataset. By typing “clear”, it specifies that it is okay to replace 

the data in memory, even though the current data have not been saved to disk 

 

2. Summary statistics  

 

sort ma co ye 

 

 

Sort command arrange the observations of the current data into ascending order based on the values of the 

variables. Data can be sorted by more than one variable, and in such cases, the sort order is lexicographic. If we 

sort the data by two variables, for instance, the data are placed in ascending order of the first variable, and 

then observations that share the same value of the first variable are placed in ascending order of the second 

variable. Here, we sort market, model, year. The data are in ascending order of market and within each market 

category, the data are in ascending order of model and within each model code, the data are in ascending order 

of year. Therefore, the oldest year of model No.1 in Belgium is 1983. 

 

 

list ma co ye qu in 1/20 

 

list displays the values of variables. Here we list first 20 observations’ market, model code, year, number of 

sales. 

 



 
 

tab ma 

tab ye  

 

tabulate produces a one-way table of frequency counts. Here we tabulate market and find there are five 

categories of market: Belgium France Germany Italy and UK. There are 2,673 cars in Belgium, which accounts 

for 23.14%. By tabulating year, we can see the oldest year is 1970 and the most recent year is 1999 in our 

dataset. 

 

 

 



gen logq = ln(qu) 

gen logp = ln(pr) 

gen logpop = ln(pop) 

gen loggdp = ln(ngdp) 

 

Then we generate new variables: logq logp logpop loggdp, which are log forms of variable: number of sales, 

prices, population and nominal GDP. 

 

 

hist logq, bin(50) 

hist logp, bin(50) 

 

A histogram is a plot that lets you discover, and show, the underlying frequency distribution (shape) of a set 

of continuous data. This allows the inspection of the data for its underlying distribution (e.g., normal 

distribution), outliers, skewness, etc. “histogram” command assumes that the variable is continuous, so you 

need to type only histogram followed by the variable name. If you add up the area of the bars, you would get 

1. 

 

  

 

 

scatter logp logq 

 

Scatter plots are important in statistics because they can show the extent of correlation, if any, between the 

values of observed quantities or phenomena (called variables). If no correlation exists between the variables, 

the points appear randomly scattered on the coordinate plane. If a large correlation exists, the points 

concentrate near a straight line. 

https://statistics.laerd.com/statistical-guides/types-of-variable.php


 
 

 

 

3. Simple regressions: isoelastic demands (CES) 

 

1) OLS 

 

reg logq logp, robust 

 

 

 

Robust here means robust(unclustered) variance estimator. This simple OLS shows the negative relationship 

between price and quantity demanded. 

 

Notes: The difference between three types of variance estimators: OLS, robust, and robust cluster is following: 

 

a) OLS variance estimator:  VOLS = s2 * (X'X)-1  

Where  s2 = (1/(N - k)) ΣNi=1 ei2     

 

b) Robust (unclustered) variance estimator: Vrob = (X'X)-1 * [ ΣNi=1 (ei*xi)' * (ei*xi) ] * (X'X)-1 

 

c) Robust cluster variance estimator: Vcluster = (X'X)-1 * Σnj=1 uj'*uj * (X'X)-1 

Where uj = Σj_cluster ei*xi  and n is the total number of clusters. 

 



Above, ei is the residual for the ith observation and xi is a row vector of predictors including the constant.  For 

simplicity, I omitted the multipliers (which are close to 1) from the formulas for Vrob and Vclusters. The formula 

for the clustered estimator is simply that of the robust (unclustered) estimator with the individual ei*xi’s 

replaced by their sums over each cluster. 

 

However, simple OLS does not consider simultaneity issue between price and unobservable attributes. Therefore 

we would like to control Fixed effects to see the difference: 

 

2) OLS controlling model Fixed Effect 

 

areg logq logp, robust a(co) 

 

 
 

Typical use of areg: “areg depvar indvar1 indvar2, absorb(groupvar)”   

areg and absorb() allows you to dummy for qualitative variables and obtain OLS regression results, except it 

does not create new variables or include coefficients for these dummies in regression results. 

 

Notes: Regression results for our variable of interest and other quantitative covariates remain identical whether 

you: 1) manually generate dummies and include them in the regression (and use "reg") or 2) use the areg 

method 

 

 

 

 

3) OLS controlling model & time Fixed Effect 

 

areg logq logp i.ye, robust a(co) 

 

Here we can see time dummies are manually generated and added into results table instead of being absorbed. 

 

Notes: areg fits a linear regression absorbing one categorical factor. areg is designed for datasets with many 

groups, but not a number of groups that increases with the sample size. xtreg, fe command for an estimator 

that handles the case in which the number of groups increases with the sample size. 

 



 

 

 

4) OLS controlling model & time & market Fixed Effect  

 

areg logq logp i.ye i.ma, robust a(co) 



 

 

 

Another command reghdfe performs exactly same as above:  

reghdfe logq logp, vce(robust) a(co ma ye) 

 

 

reghdfe is a Stata package that runs linear and instrumental-variable regressions with many levels of fixed 

effects. Within Stata, it can be viewed as a generalization of areg/xtreg, with several additional features: 

a) Supports two or more levels of fixed effects. 

b) It can estimate not only ols regressions but two-stage least squares, instrumental-variable regressions, and 

linear gmm (via the ivreg2 and ivregress commands). 



c)  Careful estimation of degrees of freedom, taking into account nesting of fixed effects within clusters, as 

well as many possible sources of collinearity within the fixed effects. 

d)  Even with only one level of fixed effects, it is faster than areg/xtreg 

 

 

From the result, we can see the endogeneity problem due to the correlation of price with time-invariant country 

heterogeneity seems much more important than the endogeneity problem due to the correlation of price with 

time-invariant model heterogeneity 

 

 

5) OLS adding population, GDP into explanatory variables, controlling model & time & market Fixed Effect 

 

reghdfe logq logp logpop loggdp, vce(robust) a(co ma ye) 

 

 

By controlling GDP and population, the coefficient of logp starts to look like a reasonable elasticity of demand. 

 

 

 

 

 

4. Construction of market shares 

 

We use population as measure of market size, consider the demand is at the household level and assume an 

average family size of 4 members. Therefore, market size H = pop/4 . This will not be very important for the 

empirical results because eventually we are going to control for market*year fixed effects. 

 

 

 



gen msize = pop/4 

 

Then we generate market share, which equals to number of sales divided by market size: 

 

gen share = qu/msize 

 

egen sum_share = sum(share), by(ma ye) 

 

One of Stata’s most powerful and useful commands is egen. Like generate, it is used to create new variables, 

but it is much more than that.  Using egen difficult and tedious variables can be created easily.  Some 

examples are variables whose values are the mean of another variable for each group such as sociability for 

males and females.  You can also use egen to create other variables that count the number of observations 

that fit a certain criteria, or even simply number observations.  Here we are producing sum of market shares, 

separately for groups defined by one or more variables specified as arguments to by(), i.e, by market and year. 

 

 

The outside good’s market share in a given country, given year, is defined as follows: 

  

gen share0 = 1 - sum_share 

 

sum share share0 

 

 

Let’s generate log of odds-ratio and make the histogram: 

 

gen lsj_ls0 = ln(share/share0) 

 

hist lsj_ls0, bin(50) 

 



 

 

 

5. Logit demand regressions (OLS and FE) 

 

Now we are ready for logit demand estimation: 

 

1) OLS 

reg lsj_ls0 logp, robust 

 
 

Remember that we need to solve the price endogeneity problem by including FEs: 

 

2) OLS with model FE 

 

reghdfe lsj_ls0 logp, vce(robust) a(co) 

 

 



3) OLS with model & year FE 

 

reghdfe lsj_ls0 logp, vce(robust) a(co ye) 

 

 

 

4) OLS with model & year & market FE 

 

reghdfe lsj_ls0 logp, vce(robust) a(co ye ma) 

 
 

 

5) OLS controlling population & GDP with model & year & market FE 

 

reghdfe lsj_ls0 logp logpop loggdp, vce(robust) a(co ma ye)  



 

 

 

6) OLS controlling population & GDP with brand & year & market FE 

 

reghdfe lsj_ls0 logp logpop loggdp, vce(robust) a(ma ye brd)  

 

 

 

7) OLS controlling population & GDP with brand & year & market FE and including model attributes: 

 

reghdfe lsj_ls0 logp sp ac li wi cy hp we pl do le he logpop loggdp, vce(robust) a(ma ye brd)  



 
 

The effect of some characteristics have the expected sign (sp ac li wi), ,i.e, maximum speed, acceleration time, 

fuel efficiency, width. The own price-elasticity of demand seems kind of reasonable. 

 


